The Logistic Population Model

```
> restart:with(plots):
```

Suppose the per capita rate of growth is b, the per capita rate of death from natural causes is d, $k = b - d$, and the per interaction death rate from intraspecies competition is k_1. Then

$$\frac{dp}{dt} = b\,p - d\,p - \frac{k_1\,p\,(p-1)}{2},$$

giving us the **logistic model**

$$\frac{dp}{dt} = k\,p - \frac{k_1\,p\,(p-1)}{2},\quad p(0) = p_0 \text{ or}$$

.

$$\frac{dp}{dt} = -Ap\left(p - p_1\right)$$

where

$$A = \frac{k_1}{2} \text{ and } p_1 = \frac{2 \cdot k}{k_1} + 1\,.$$

We enter the model.

```
> deq:=diff(p(t),t)=-A*p(t)*(p(t)-p[1]);
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}t}\,p(t) = -A\,p(t)\,\left(p(t) - p_1\right)$$

```
> IC:=p(0)=p[0];
```

$$IC := p(0) = p_0$$

Now we find the solution.

```
> soln:=dsolve({deq,IC},p(t));
```

$$soln := p(t) = -\frac{p_1\,p_0}{\mathrm{e}^{-A\,p_1\,t}\,p_0 - \mathrm{e}^{-A\,p_1\,t}\,p_1 - p_0}$$

Let's simplify this solution and turn it into a function.

```
> P:=simplify(rhs(soln),t);
```

$$P := -\frac{p_1\,p_0}{\left(p_0 - p_1\right)\,\mathrm{e}^{-A\,p_1\,t} - p_0}$$

Let's choose values for A, p_0 and p_1 with $p_0 < p_1$.

```
> A:=.2;p[0]:=3;p[1]:=6;
```

$$A := 0.2$$

$$p_0 := 3$$

$$p_1 := 6$$

Let's look at our function now.

```
> P:=simplify(rhs(soln),t);
```

$$P := \frac{6}{e^{-1.2\,t}+1}$$

Let's graph the solution along with the horizontal line $p_1 = 6$.

```
> p1:=plot(P,t=-5..5,view=[-5..5,0..8]):
  p2:=plot(6,t=-5..5,view=[-5..5,0..8],color=green):
  display(p1,p2);
```



Next let's suppose $p_0 > p_1$. So take $p_0 = 6$ and $p_1 = 4$.

```
> p[0]:=6;p[1]:=4;
```

$$p_0 := 6$$

$$p_1 := 4$$

Let's look at our function now.

```
> P:=simplify(rhs(soln),t);
```

$$P := -\frac{12}{e^{-0.8\,t}-3}$$

Let's graph this solution along with the horizontal line $p_1 = 4$.

```
> p3:=plot(P,t=-5..5,view=[-5..5,0..8],discont=true):
  p4:=plot(4,t=-5..5,view=[-5..5,0..8]):
  display(p3,p4);
```



Notice that in each case the population approaches p_1 , the **carrying capacity of the environment**.

$p_0 = 0$ and $p_0 = p_1$ are called **equilibrium populations**.